# Maintainer: Felix Yan <felixonmars@archlinux.org>

_hkgname=should-not-typecheck
pkgname=haskell-should-not-typecheck
pkgver=2.1.0
pkgrel=240
pkgdesc="A HUnit/hspec assertion library to verify that an expression does not typecheck"
url="https://github.com/CRogers/should-not-typecheck"
license=('BSD')
arch=('loong64' 'x86_64')
depends=('ghc-libs' 'haskell-hunit')
makedepends=('ghc' 'haskell-hspec' 'haskell-hspec-expectations')
source=(https://hackage.haskell.org/packages/archive/${_hkgname}/${pkgver}/${_hkgname}-${pkgver}.tar.gz)
sha512sums=('46ba4435a4408e23dfe6e04f827a7efcfd3c94a4a53a0562061d5cd40c11fcbae7569e695329fe98f9d9ecea1999ac68da6594d2ca60c2efccdac2d6d674d65d')

build() {
    cd $_hkgname-$pkgver

    runhaskell Setup configure -O --enable-shared --enable-executable-dynamic --disable-library-vanilla \
        --prefix=/usr --docdir=/usr/share/doc/$pkgname --datasubdir=$pkgname --enable-tests \
        --dynlibdir=/usr/lib --libsubdir=\$compiler/site-local/\$pkgid --ghc-option=-fllvm
    runhaskell Setup build $MAKEFLAGS
    runhaskell Setup register --gen-script
    runhaskell Setup unregister --gen-script
    sed -i -r -e "s|ghc-pkg.*update[^ ]* |&'--force' |" register.sh
    sed -i -r -e "s|ghc-pkg.*unregister[^ ]* |&'--force' |" unregister.sh
}

check() {
    cd $_hkgname-$pkgver
    runhaskell Setup test --show-details=direct
}

package() {
    cd $_hkgname-$pkgver

    install -D -m744 register.sh "$pkgdir"/usr/share/haskell/register/$pkgname.sh
    install -D -m744 unregister.sh "$pkgdir"/usr/share/haskell/unregister/$pkgname.sh
    runhaskell Setup copy --destdir="$pkgdir"
    install -D -m644 LICENSE "$pkgdir"/usr/share/licenses/$pkgname/LICENSE
    rm -f "$pkgdir"/usr/share/doc/$pkgname/LICENSE
}
